CoW-AMM Standalone Audit

This document presents the finding of a smart contract audit conducted by Côme du Crest for Gnosis.

Scope

The scope includes all contracts within CoW-amm. As the team is pertinacious about security, the contracts have been reviewed multiple times as the code evolved. In historical order, the code has been reviewed as of commits d673529, 145384c, and 4615cf8. Last fixes have been added in commit 5718abc.

The scope includes all contracts within the `cow-amm/src` directory.

Context

Previously, the AMM took the form of a Safe contract that used the `SignatureVerifierMuxer` and `ComposableCow` to verify signatures on GPv2 orders. The contract logic has been turned into a standalone contract `ConstantProduct.sol` that each user can deploy to hold their funds and act as the AMM that verifies signature using EIP1271.

Status

The audit has been sent to the developer team.

Issues

▼ [Info] [d673529] [Acknowledged] Unenforced minimum trade amount

Summary

The code to generate the trade order attempts to enforce a minimum trade amount, however this amount is not enforced during signature verification and orders below minimum trade amount will still be accepted. Additionally, orders can be partially fillable in which case the minimum trade amount will not be respected anyway.

Vulnerability Detail

The function to get the GPv2 order from the AMM reverts when trade amount in token0 is below a minimum trade amount:

```
    function _getTradeableOrder(address owner, bytes calldata staticInput)
        internal
        view
        returns (GPv2Order.Data memory order)
    {
                ...

                if (tradedAmountToken0 < data.minTradedToken0) {
            revert IWatchtowerCustomErrors.PollTryAtEpoch(
                Utils.validToBucket(MAX_ORDER_DURATION) + 1, "traded amount too small"
            );
          }

                ...
        }
```

The value of `data.minTradedToken0` is user-provided anyway and there is no verification on it (unless called through `ComposableCoW.getTradeableOrderWithSignature()`).

No rules enforce this is the case during order verification.

Impact

The function to get the tradeable order is not meant to be called on-chain and is just a helper function, I understand there is a trust assumption that most of the time the orders executed from the AMM will be computed from

`getTradeableOrder()`. However, developers should be aware that this value is not strictly enforced and any forged order could execute below this.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/d673529e3d5df028aca0eab23db90b5c8790e0ff/src/ConstantProduct.sol#L131-L135

Recommendation

Acknowledge that this is intended behaviour.

Status

Acknowledged.

▼ [Info] [d673529] [Acknowledged] Unenforced order expiry time

Summary

The `_verify()` function attempts to enforce that order's validity in terms of max duration but the `order.validTo` field is user-provided and does not prevent old orders from executing.

Vulnerability Detail

The `_verify()` function reverts when `order.validTo` is too far in the future:

```
// We add a maximum duration to avoid spamming the orderbook and force
// an order refresh if the order is old.
if (order.validTo > block.timestamp + MAX_ORDER_DURATION) {
    revert IConditionalOrder.OrderNotValid("validity too far in the future");
}
```

However, the `order.validTo` can be a forged value by the order provider. If not a forged value, then this condition can never be met as `_getTradeableOrder()` computes the `order.validTo` value as `Utils.validToBucket(MAX_ORDER_DURATION)`.

Impact

This check does not seem to force a refresh of old orders or prevent `order.validTo` manipulation.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/d673529e3d5df028aca0eab23db90b5c8790e0ff/src/ConstantProduct.sol#L202-L206

https://github.com/cowprotocol/cow-amm/blob/d673529e3d5df028aca0eab23db90b5c8790e0ff/src/ConstantProduct.sol#L143

Recommendation

Delete the check.

Status

Acknowledged.

▼ [Info] [d673529] [Acknowledged] Incorrect rounding in _getTradeableOrder()

Summary

The documentation states that they want to round down the sell amount and round up the buy amount, which makes sens to guarantee the interests of the AMM. However, the rounding of the buyAmount may result in a rounding down from the exact calculation.

Vulnerability Detail

The `sellAmount` is properly rounded down from the calculation:

```
sellAmount = selfReserve0 / 2 - Math.ceilDiv(selfReserve1TimesPriceNumerator, 2 * priceDen
```

However, the `buyAmount` calculation re-uses the `sellAmount` value which has been rounded down and rounds up the final result:

```
buyAmount = Math.mulDiv(
    sellAmount,
    selfReserve1TimesPriceNumerator + (priceDenominator * sellAmount),
    priceNumerator * selfReserve0,
    Math.Rounding.Up
);
```

There is no guarantee that when we express `buyAmount` in terms of reserves and oracle price the returned value is a rounded up value of the calculation since an intermediate rounded down value is used.

Impact

If we consider that the buyAmount shall be expressed in terms of buyAmount and not reserve and oracle price, then the rounding issue is irrelevant. Otherwise, since the calculation should not give the best available price for the AMM, the impact should be insignificant.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/d673529e3d5df028aca0eab23db90b5c8790e0ff/src/ConstantProduct.sol#L107-L129

Recommendation

Acknowledge the issue.

Status

Acknowledged.

▼ [Low] [145384c] [Acknowledged] DOS orders on empty commitments by sending tokens

Summary

If CoW Protocol solvers do not commit on a hash and the commitment in the AMM is equal to `bytes32(0)`, the complete order parameters will be checked against the AMM's returned order via `_getTradeableOrder()` which relies on the internal balance of the AMM for computing the order. Anyone can make the signature verification fail by sending a few wei of tokens to the AMM in a front-running transaction.

Vulnerability Detail

The signature verification process starts with verifying :

```
function _verify(address owner, bytes32 orderHash, bytes calldata staticInput, GPv2Order.D
        internal
        view
    {
        requireMatchingCommitment(owner, orderHash, staticInput, order);
        ...
    }
```

`requireMatchingCommitment()` checks that either the committed hash matches the order hash or if the committed hash is `bytes32(0)` then it checks that the order matches what is returned by `_getTradeableOrder()`:

```
    function requireMatchingCommitment(
        address owner,
        bytes32 orderHash,
        bytes calldata staticInput,
        GPv2Order.Data calldata order
```

```
    ) internal view {
        bytes32 committedOrderHash = commitment[owner];
        if (orderHash != committedOrderHash) {
            if (committedOrderHash != EMPTY_COMMITMENT) {
                revert OrderDoesNotMatchCommitmentHash();
            }
            GPv2Order.Data memory computedOrder = _getTradeableOrder(owner, staticInput);
            if (!matchFreeOrderParams(order, computedOrder)) {
                revert OrderDoesNotMatchDefaultTradeableOrder();
            }
        }
    }

    function matchFreeOrderParams(GPv2Order.Data calldata lhs, GPv2Order.Data memory rhs)
        internal
        pure
        returns (bool)
    {
        bool sameSellToken = lhs.sellToken == rhs.sellToken;
        bool sameBuyToken = lhs.buyToken == rhs.buyToken;
        bool sameSellAmount = lhs.sellAmount == rhs.sellAmount;
        bool sameBuyAmount = lhs.buyAmount == rhs.buyAmount;
        bool sameValidTo = lhs.validTo == rhs.validTo;
        bool sameKind = lhs.kind == rhs.kind;
        bool samePartiallyFillable = lhs.partiallyFillable == rhs.partiallyFillable;

        // The following parameters are untested:
        // - receiver
        // - appData
        // - feeAmount
        // - sellTokenBalance
        // - buyTokenBalance

        return sameSellToken && sameBuyToken && sameSellAmount && sameBuyAmount && sameVal
            && samePartiallyFillable;
    }
```

However the exact values for sell and buy amounts can easily be influenced by sending a few wei of the sell or buy tokens to the AMM which will result in the verification failing with `OrderDoesNotMatchDefaultTradeableOrder`.

Impact

Anyone can DOS an order dependant on an empty commitment by sending a few wei of tokens to the AMM. If CoW protocol solvers rely on this behaviour, they can invalidate each other's batches with front-running transactions.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/145384cca64c44dd36bf68ae874f172b302eba10/src/ConstantProduct.sol#L245-L249

https://github.com/cowprotocol/cow-amm/blob/145384cca64c44dd36bf68ae874f172b302eba10/src/ConstantProduct.sol#L344-L392

https://github.com/cowprotocol/cow-amm/blob/145384cca64c44dd36bf68ae874f172b302eba10/src/ConstantProduct.sol#L174-L180

Recommendation

Solvers should not rely on the empty commitment behaviour and should commit to the order they execute at the beginning of a batch. As far as I understood from the protocol team, this is the behaviour they expect solvers to apply in the long run. The empty commitment feature is here to help solvers discover the use of the AMM before they adapt to it.

If that is indeed the case, no fixing is required.

Status

Acknowledged.

▼ [Low] [145384c] [Acknowledged] Solvers can DOS orders by committing to different hash

Summary

CoW Protocol solvers that rely on a specific committed to hash before a batch execution to get their order validated can be denied by other solvers committing to any other hash in a front-running transaction. This is specifically the case for solvers relying on the empty commitment behaviour.

Vulnerability Detail

The signature verification process starts with verifying :

```
function _verify(address owner, bytes32 orderHash, bytes calldata staticInput, GPv2Order.D
        internal
        view
    {
        requireMatchingCommitment(owner, orderHash, staticInput, order);
        ...
    }
```

`requireMatchingCommitment()` checks that either the committed hash matches the order hash or if the committed hash is `bytes32(0)` then it checks that the order matches what is returned by `_getTradeableOrder()`:

```
    function requireMatchingCommitment(
        address owner,
        bytes32 orderHash,
        bytes calldata staticInput,
        GPv2Order.Data calldata order
    ) internal view {
        bytes32 committedOrderHash = commitment[owner];
        if (orderHash != committedOrderHash) {
            if (committedOrderHash != EMPTY_COMMITMENT) {
                revert OrderDoesNotMatchCommitmentHash();
            }
            GPv2Order.Data memory computedOrder = _getTradeableOrder(owner, staticInput);
            if (!matchFreeOrderParams(order, computedOrder)) {
                revert OrderDoesNotMatchDefaultTradeableOrder();
            }
        }
    }
```

If any adversary CoW Protocol solvers commits to a hash before submission of a batch relying on an empty commitment or specific hash, then the signature will not be verified and the order invalid.

Impact

Any solver can front-run a batch execution relying on a committed hash and not committing in the same transaction. Due to the order hash not being cleared after signature verification (because signature verification legitimately uses a view function), any solver committing to any value will permanently deny other solvers relying on the empty commitment feature until they re-apply a `bytes32(0)` commitment.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/145384cca64c44dd36bf68ae874f172b302eba10/src/ConstantProduct.sol#L245-L249

https://github.com/cowprotocol/cow-amm/blob/145384cca64c44dd36bf68ae874f172b302eba10/src/ConstantProduct.sol#L344-L360

Recommendation

The protocol team already expressed interests to use `TSTORE` and `TLOAD` in the future to alleviate the issue. In the meantime, the committed hash could be bound to each independent solver by using a mapping `commitment[owner][solver] => hash` instead of `commitment[owner] => hash`. This would require to find out the solver's address in the signature verification process either via additional static input data, via the use of `tx.origin`, or by providing it in `payload.offchainInput`.

The protocol team may acknowledge this issue.

Status

Acknowledged.

▼ [Low] [145384c] [Acknowledged] Balancer's oracle vulnerable to manipulation

Summary

Balancer's functions `vault.getPoolTokens(poolId)` is vulnerable to a read-only re-entrancy attack that allows manipulating the returned values of token balances. This means that the price returned by the Balancer oracle can be arbitrarily manipulated.

See: Balancer's forum and provided library solution.

Vulnerability Detail

The `getPrice()` function of the `BalancerWeightedPoolPriceOracle` uses `vault.getPoolTokens(poolId)` to get the balances of each token in the pool and relies on these balances to compute the price:

```
    function getPrice(address token0, address token1, bytes calldata data) external view r
        IWeightedPool pool;
        IERC20[] memory tokens;
        uint256[] memory balances;
        uint256[] memory weights;
        {
            // Note: function calls in this scope aren't affected by the paused
            // state
            bytes32 poolId = abi.decode(data, (Data)).poolId;
            // If the pool isn't registered, then the next call reverts
            try vault.getPool(poolId) returns (address a, IVault.PoolSpecialization) {
                pool = IWeightedPool(a);
            } catch (bytes memory) {
                revert IConditionalOrder.OrderNotValid("invalid pool id");
            }

            (tokens, balances,) = vault.getPoolTokens(poolId);

            ...
        }
        ...

        uint256 priceNumerator = balanceToken0 * weightToken1;
        uint256 priceDenominator = balanceToken1 * weightToken0;

        ...
    }
```

The value of balances can be manipulated via read-only re-entrancy impacting the computed price.

Impact

The manipulated price impacts the order computed from `getTradeableOrder()` by the AMM. This in turns impact signature verification that relies on the empty commitment feature for matching the executed order to the computed order by the AMM.

However, it does not impact the logic of the AMM for verifying that it should accept an order based on its underlying curve as defined in `_verify()`.

No loss of funds should be caused by this issue.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/145384cca64c44dd36bf68ae874f172b302eba10/src/oracles/BalancerWeightedPoolPriceOracle.sol#L53-L117

Recommendation

It is not possible to use Balancer's recommanded library to alleviate the read-only re-entrancy because it is not a view function. Ultimately, I understand the oracle is here to help CoW Protocol solvers to produce valid orders rather than enforce order validity.

Since no loss of funds is at stake, the protocol team may acknowledge the issue.

Status

Acknowledged.

▼ [Info] [4615cf8] [Acknowledged] AMM Owner can DOS orders

Summary

Owners of AMM can DOS orders by withdrawing / depositing / updating the AMM parameters / disabling trading to invalidated a previously valid order. They can also create AMM that never validate signatures for default commitments using a reverting oracle.

Vulnerability Detail

AMM owners can for example disable or enable trading at will:

```
    function updateParameters(
        ConstantProduct amm,
        uint256 minTradedToken0,
        IPriceOracle priceOracle,
        bytes calldata priceOracleData,
        bytes32 appData
    ) external onlyOwner(amm) {
        ConstantProduct.TradingParams memory data = ConstantProduct.TradingParams({
            minTradedToken0: minTradedToken0,
            priceOracle: priceOracle,
            priceOracleData: priceOracleData,
            appData: appData
        });
        _disableTrading(amm);
        _enableTrading(amm, data);
    }

    function disableTrading(ConstantProduct amm) external onlyOwner(amm) {
        _disableTrading(amm);
    }
```

Using this function they can update the state of the AMM so that previously valid order in the order book become invalid.

They can also use a custom oracle that always reverts to create orders that are never valid for empty commitments:

```
    function getTradeableOrder(TradingParams memory tradingParams) public view returns (GP
        (uint256 priceNumerator, uint256 priceDenominator) =
            tradingParams.priceOracle.getPrice(address(token0), address(token1), tradingPa
```

```
        ...
    }
```

Impact

If orders created by AMMs deployed by `ConstantProductFactory` are automatically picked up and added to the order book off-chain, the order book can be polutted by many invalid orders. Non-trusted owners of AMMs can also selectively DOS orders using a custom oracle or simply by enabling / disabling the AMM when they wish.

Code Snippets

https://github.com/cowprotocol/cow-amm/blob/4615cf8b80786e4de1d76cd1ad41f69353c0b797/src/ConstantProduct.sol#L245

https://github.com/cowprotocol/cow-amm/blob/4615cf8b80786e4de1d76cd1ad41f69353c0b797/src/ConstantProductFactory.sol#L135-L137

Recommendation

Acknowledge the issue. Otherwise, add a delay to functions that update the AMM state (deposit / withdrawal / enabling / disabling, etc ...). Note that the owner can still deposit tokens freely by directly sending the tokens to the AMM.

Status

Acknowledged.

▼ Optimisations and miscellaneous

Summary

This part lists minor gas/code optimizations that shouldn't make the code less readable or improve overall readability. It also lists questions about unclear code segments.

[4615cf8] [Fixed] Save gas by moving order of operations

The function to verify the input order of the AMM may save gas by moving the allocation of sell / buy reserves after the inversion of sell / buy token:

```
    function verify(TradingParams memory tradingParams, GPv2Order.Data memory order) publi
        IERC20 sellToken = token0;
        IERC20 buyToken = token1;
-       uint256 sellReserve = sellToken.balanceOf(address(this));
-       uint256 buyReserve = buyToken.balanceOf(address(this));
        if (order.sellToken != sellToken) {
            if (order.sellToken != buyToken) {
                revert IConditionalOrder.OrderNotValid("invalid sell token");
            }
            (sellToken, buyToken) = (buyToken, sellToken);
-           (sellReserve, buyReserve) = (buyReserve, sellReserve);
        }
        if (order.buyToken != buyToken) {
            revert IConditionalOrder.OrderNotValid("invalid buy token");
        }
+       uint256 sellReserve = sellToken.balanceOf(address(this));
+       uint256 buyReserve = buyToken.balanceOf(address(this));
        ...
    }
```

Conclusion

No important security issues have been found during the audits.